# $Id: PKGBUILD 79005 2012-10-28 18:45:57Z jlichtblau $
# Maintainer: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Geoffroy Carrier <geoffroy.carrier@aur.archlinux.org>
# Contributor: Marco Bartholomew <marco@marcobartholomew.com>

pkgname=monit
pkgver=5.5
pkgrel=2
pkgdesc="Utility for managing and monitoring, processes, files, directories and devices on a *NIX system"
arch=('i686' 'x86_64')
url="http://mmonit.com/monit/"
license=('GPL3')
depends=('openssl')
backup=('etc/monitrc')
changelog=$pkgname.changelog
source=(http://mmonit.com/$pkgname/dist/$pkgname-$pkgver.tar.gz \
        monitd monit.service)
sha256sums=('8276b060b3f0e6453c9748d421dec044ddae09d3e4c4666e13472aab294d7c53'
            'b235b2fdeec49015d6b74f028d161b289663692f9ee2d2d78cf52d8db87ca496'
            'cf607209e9e5d39e708152dc181525570371fab750f828e890e2b6a00b81ebc4')

build() {
  cd ${srcdir}/$pkgname-$pkgver

  ./configure --prefix=/usr

  make
}

package() {
  cd ${srcdir}/$pkgname-$pkgver

  make DESTDIR=${pkgdir} install

  install -D -m755 ${srcdir}/monitd  ${pkgdir}/etc/rc.d/monitd
  install -D -m700 monitrc ${pkgdir}/etc/monitrc
  
  install -D -m644 ${srcdir}/$pkgname.service ${pkgdir}/usr/lib/systemd/system/$pkgname.service
}
